

EN 2111 – Electronic Circuit Design

UART Implementation in FPGA



Department of Electronic and Telecommunication Engineering
University of Moratuwa, Sri Lanka

Group 33

Submitted by

| | |
|---------------------|---------|
| Wickramasinghe S.D. | 220701X |
| Weragoda W.A.A.P. | 220692R |
| Wijenayaka M.B.T.I. | 220711D |

Submitted on May 22, 2025

Contents

| | | |
|----------|-----------------------------------|-----------|
| 1 | Introduction | 2 |
| 2 | UART Transceiver Design | 3 |
| 2.1 | UART Configuration | 3 |
| 2.2 | UART Transmitter | 3 |
| 2.3 | UART Receiver | 4 |
| 3 | Testbench Development | 7 |
| 3.1 | Testbench Configuration | 8 |
| 4 | ModelSim Simulation | 10 |

1 Introduction

UART, or Universal Asynchronous Receiver–Transmitter, is one of the world’s most-used hardware communication protocols. It is a serial, full-duplex protocol that can be configured to operate at various baud rates. “Asynchronous” means there is no shared clock signal on the transmission medium; instead, both ends must agree on the same baud rate to correctly sample the bit stream.

In a typical UART design, the Transmitter (Tx) and Receiver (Rx) are implemented as separate modules. The Tx converts parallel data from a host bus into a serial bit stream, inserting start, stop, and optional parity bits. The Rx detects framing boundaries in the incoming serial stream, samples each bit at the configured baud rate, and reconstructs the original parallel word.

- **Transmitter (Tx):** Encodes parallel data into a serial format with start/stop bits.
- **Receiver (Rx):** Detects start bit, samples data bits, checks stop bit, and outputs parallel data.




Figure 1: Conventional UART with separate Tx and Rx connected to a data bus

In this assignment, we go beyond the conventional split design by integrating both transmitter and receiver into a single *transceiver* module. This unified approach shares clock-division logic, control signals, and I/O buffers, thereby reducing resource utilization on the FPGA and simplifying timing coordination between Tx and Rx paths.

The objectives of this work are to:

1. Develop a Verilog RTL implementation of a fully integrated UART transceiver
2. Create a comprehensive testbench using Verilog, with waveform inspection in ModelSim, to verify protocol compliance, timing accuracy, and error detection.
3. Synthesize and deploy the transceiver design onto a DE0-Nano FPGA board, perform hardware-in-the-loop validation with peer devices, and capture real-time serial waveforms on an oscilloscope.

By combining Tx and Rx into a single RTL block, this design not only conserves FPGA logic resources but also ensures tight synchronization and simpler integration into larger serial-communication subsystems.

2 UART Transceiver Design

2.1 UART Configuration

This UART transceiver uses a common baud-rate generator and I/O buffering for both TX and RX. The configuration parameters are:

- System clock: `clk`
- Reset: active-low `rst_n`
- Baud Rate: defined by

$$\text{CLKS_PER_BIT} = \frac{\text{CLK_FREQ}}{\text{BAUD_RATE}} = \frac{50\,000\,000}{115\,200}$$

- Default parameters:
 - `CLK_FREQ` = 50 000 000 (50 MHz)
 - `BAUD_RATE` = 115 200 bps

2.2 UART Transmitter

The transmitter FSM has four states controlling the serial output `tx` and busy flag `tx_busy`:

- IDLE:
 - Drive `tx` = 1 (line idle)
 - Clear `tx_clk_count` and `tx_bit_count`
 - Wait for `tx_start` && `!tx_busy` to latch `tx_data` into `tx_shift_reg` and assert `tx_busy`
- START:
 - Drive `tx` = 0 for one bit period
 - Increment `tx_clk_count` until it reaches `CLKS_PER_BIT - 1`, then reset counter and move to DATA
- DATA:
 - Drive `tx` = `tx_shift_reg[0]` (LSB first)
 - On each full bit period, shift `tx_shift_reg` right by one and increment `tx_bit_count`
 - After 8 bits, transition to STOP
- STOP:
 - Drive `tx` = 1 for one bit period
 - At end of period, deassert `tx_busy` and return to IDLE

2.3 UART Receiver

The receiver FSM samples the synchronized input `rx_d2` at mid-bit times and assembles bytes:

- **Input Synchronization:**
 - Two-flip-flop synchronizer (`rx_d1`, `rx_d2`) to avoid metastability
- **IDLE:**
 - Wait for `rx_d2 == 0` (start-bit falling edge)
 - Reset `rx_clk_count` and `rx_bit_count`
- **START:**
 - Count to midpoint: `CLKS_PER_BIT - 12`
 - Verify `rx_d2` still low; if valid, wait remainder of bit period then go to **DATA**
 - On false start, return to **IDLE**
- **DATA:**
 - At each mid-bit point, sample `rx_d2` into the MSB of `rx_shift_reg` and shift existing bits down
 - After 8 bits, transition to **STOP**
- **STOP:**
 - Wait one full bit period, then assert `rx_done` for one clock and transfer `rx_shift_reg` to `rx_data`
 - Return to **IDLE**

```
1 module uart_transceiver(  
2     input wire      clk,          // System clock  
3     input wire      rst_n,        // Active low reset  
4     // UART TX Interface  
5     input wire      tx_start,     // Start transmission  
6     input wire [7:0] tx_data,     // Data to transmit  
7     output reg      tx_busy,      // Transmitter busy flag  
8     output reg      tx,           // Serial TX output  
9     // UART RX Interface  
10    input wire      rx,           // Serial RX input  
11    output reg      rx_done,      // Reception complete flag  
12    output reg [7:0] rx_data      // Received data byte  
13 );  
14  
15 // Parameters  
16 parameter CLK_FREQ    = 50_000_000; // 50 MHz system clock  
17 parameter BAUD_RATE    = 115200;    // Desired baud rate  
18 localparam CLKS_PER_BIT = CLK_FREQ / BAUD_RATE;
```

```

19
20 // State definitions
21 localparam IDLE = 2'b00;
22 localparam START = 2'b01;
23 localparam DATA = 2'b10;
24 localparam STOP = 2'b11;
25
26 // TX registers
27 reg [1:0] tx_state;
28 reg [15:0] tx_clk_count;
29 reg [2:0] tx_bit_count;
30 reg [7:0] tx_shift_reg;
31
32 // RX registers
33 reg [1:0] rx_state;
34 reg [15:0] rx_clk_count;
35 reg [2:0] rx_bit_count;
36 reg [7:0] rx_shift_reg;
37 reg rx_d1, rx_d2; // Double-flop synchronizer
38
39 // Synchronize RX
40 always @(posedge clk or negedge rst_n) begin
41     if (!rst_n) begin
42         rx_d1 <= 1'b1;
43         rx_d2 <= 1'b1;
44     end else begin
45         rx_d1 <= rx;
46         rx_d2 <= rx_d1;
47     end
48 end
49
50 // TX State Machine
51 always @(posedge clk or negedge rst_n) begin
52     if (!rst_n) begin
53         tx_state <= IDLE;
54         tx_clk_count <= 0;
55         tx_bit_count <= 0;
56         tx_busy <= 1'b0;
57         tx <= 1'b1;
58         tx_shift_reg <= 8'h00;
59     end else begin
60         case (tx_state)
61             IDLE: begin
62                 tx <= 1'b1;
63                 tx_clk_count <= 0;
64                 tx_bit_count <= 0;
65                 if (tx_start && !tx_busy) begin
66                     tx_busy <= 1'b1;
67                     tx_shift_reg <= tx_data;
68                     tx_state <= START;

```

```

69         end
70     end
71     START: begin
72         tx <= 1'b0;
73         if (tx_clk_count < CLKS_PER_BIT-1) tx_clk_count <= tx_clk_count + 1;
74     else begin
75         tx_clk_count <= 0;
76         tx_state <= DATA;
77     end
78 end
79 DATA: begin
80     tx <= tx_shift_reg[0];
81     if (tx_clk_count < CLKS_PER_BIT-1) tx_clk_count <= tx_clk_count + 1;
82 else begin
83     tx_clk_count <= 0;
84     tx_shift_reg <= {1'b0, tx_shift_reg[7:1]};
85     if (tx_bit_count < 7) tx_bit_count <= tx_bit_count + 1;
86 else begin
87     tx_bit_count <= 0;
88     tx_state <= STOP;
89 end
90 end
91 end
92 STOP: begin
93     tx <= 1'b1;
94     if (tx_clk_count < CLKS_PER_BIT-1) tx_clk_count <= tx_clk_count + 1;
95 else begin
96     tx_clk_count <= 0;
97     tx_busy <= 1'b0;
98     tx_state <= IDLE;
99 end
100 end
101 endcase
102 end
103 end
104
105 // RX State Machine
106 always @(posedge clk or negedge rst_n) begin
107     if (!rst_n) begin
108         rx_state <= IDLE;
109         rx_clk_count <= 0;
110         rx_bit_count <= 0;
111         rx_done <= 1'b0;
112         rx_data <= 8'h00;
113         rx_shift_reg <= 8'h00;
114     end else begin
115         if (rx_done) rx_done <= 1'b0;
116         case (rx_state)
117             IDLE: begin
118                 rx_clk_count <= 0;

```

```

119         rx_bit_count <= 0;
120         if (rx_d2 == 1'b0) rx_state <= START;
121     end
122     START: begin
123         if (rx_clk_count == (CLKS_PER_BIT-1)/2) begin
124             if (rx_d2 == 1'b0) rx_clk_count <= rx_clk_count + 1;
125             else rx_state <= IDLE;
126         end else if (rx_clk_count < CLKS_PER_BIT-1) rx_clk_count <= rx_clk_count + 1;
127         else begin
128             rx_clk_count <= 0;
129             rx_state <= DATA;
130         end
131     end
132     DATA: begin
133         if (rx_clk_count == (CLKS_PER_BIT-1)/2) begin
134             rx_shift_reg <= {rx_d2, rx_shift_reg[7:1]};
135             rx_clk_count <= rx_clk_count + 1;
136         end else if (rx_clk_count < CLKS_PER_BIT-1) rx_clk_count <= rx_clk_count + 1;
137         else begin
138             rx_clk_count <= 0;
139             if (rx_bit_count < 7) rx_bit_count <= rx_bit_count + 1;
140             else begin
141                 rx_bit_count <= 0;
142                 rx_state <= STOP;
143             end
144         end
145     end
146     STOP: begin
147         if (rx_clk_count < CLKS_PER_BIT-1) rx_clk_count <= rx_clk_count + 1;
148         else begin
149             rx_done <= 1'b1;
150             rx_data <= rx_shift_reg;
151             rx_state <= IDLE;
152         end
153     end
154 endcase
155 end
156 end
157 endmodule

```

Listing 1: UART Transceiver Snippet

3 Testbench Development

To verify the UART implementation, a Verilog testbench (`uart_tb`) was written to simulate end-to-end loopback transmission and reception of ten predefined byte patterns.

3.1 Testbench Configuration

The testbench instantiates the DUT with:

- `.CLK_FREQ = 50_000_000` and `.BAUD_RATE = 115200`.
- `tx_out` connected to `rx` via a loopback wire.
- A 50 MHz clock generated by toggling `clk` every 10 ns.
- `rst_n` pulsed at time zero to reset the DUT.
- An array of ten test vectors (`test_data[0:9]`).

During simulation, each byte is sent when `tx_busy` is low and `tx_start` is pulsed; reception is signaled by `rx_done`, and `rx_data` is compared to `tx_data`. Any mismatch reports a failure.

```
1  `timescale 1ns/1ps
2  module uart_tb();
3
4      // Parameters
5      parameter CLK_PERIOD = 20;           // 50 MHz clock
6      parameter BAUD_RATE  = 115200;
7      parameter BIT_PERIOD = 1000000000 / BAUD_RATE;
8
9      // Clock and Reset
10     reg clk    = 0;
11     reg rst_n = 0;
12
13     // Transmitter Signals
14     reg      tx_start = 0;
15     reg [7:0] tx_data  = 8'd0;
16     wire     tx_busy;
17     wire     tx_out;
18
19     // Receiver Signals
20     wire     rx_done;
21     wire [7:0] rx_data;
22
23     // Loopback connection
24     wire rx_in = tx_out;
25
26     // Test Data
27     reg [7:0] test_data [0:9];
28     integer i;
29
30     // Device Under Test
31     uart_transceiver #(
32         .CLK_FREQ (50_000_000),
33         .BAUD_RATE(BAUD_RATE)
34     ) dut (
```

```

35     .clk      (clk),
36     .rst_n    (rst_n),
37     .tx_start (tx_start),
38     .tx_data  (tx_data),
39     .tx_busy  (tx_busy),
40     .tx       (tx_out),
41     .rx       (rx_in),
42     .rx_done  (rx_done),
43     .rx_data  (rx_data)
44 );
45
46 // Clock Generator
47 always #(CLK_PERIOD/2) clk = ~clk;
48
49 // Waveform Dump
50 initial begin
51     $dumpfile("uart_transceiver_tb.vcd");
52     $dumpvars(0, uart_tb);
53 end
54
55 // Monitor Received Data
56 always @(posedge rx_done) begin
57     $display("Time %0t: Received 0x%h", $time, rx_data);
58     if (rx_data == tx_data)
59         $display("Status: PASS");
60     else
61         $display("Status: FAIL - Expected: 0x%h, Got: 0x%h", tx_data, rx_data);
62 end
63
64 // Test Procedure
65 initial begin
66     // Initialize test data
67     test_data[0] = 8'h55; test_data[1] = 8'hAA;
68     test_data[2] = 8'h00; test_data[3] = 8'hFF;
69     test_data[4] = 8'h01; test_data[5] = 8'h80;
70     test_data[6] = 8'h33; test_data[7] = 8'hCC;
71     test_data[8] = 8'hA5; test_data[9] = 8'h5A;
72
73     // Apply reset
74     #100; rst_n = 1; #100;
75
76     // Loop through test vectors
77     for (i = 0; i < 10; i = i + 1) begin
78         wait (!tx_busy);
79         @(posedge clk);
80
81         tx_data = test_data[i];
82         tx_start = 1;
83         $display("\nTime %0t: Sending 0x%h", $time, tx_data);
84

```

```

85     @(posedge clk);
86     tx_start = 0;
87
88     wait (rx_done);
89     #100;
90 end
91
92 // Done
93 #5000;
94 $display("\nUART Transceiver Test Complete");
95 $finish;
96 end
97
98 endmodule

```

Listing 2: Testbench Snippet

4 ModelSim Simulation

The UART transceiver was validated in ModelSim using a 50 MHz clock and a 115 200 baud loopback test. Figure 2 shows the captured waveform for a sequence of transmitted bytes. Each 10-bit frame (1 start bit, 8 data bits LSB-first, 1 stop bit) is sent and immediately received, confirming correct timing, bit ordering, and data integrity across all test vectors.




Figure 2: ModelSim waveform of UART loopback at 115 200 baud

Thank you!